**graph twoway fpfitci** — Twoway fractional-polynomial prediction plots with CIs

### Description

twoway fpfitci calculates the prediction for *yvar* from estimation of a fractional polynomial of *xvar* and plots the resulting curve along with the confidence interval of the mean.

### Quick start

Fractional-polynomial prediction plot of *y* on *x* with 95% confidence interval

```
twoway fpfitci y1 x
```

As above, with a 90% confidence interval

```
twoway fpfitci y1 x, level(90)
```

As above, but estimate fractional polynomial of degree 4

```
twoway fpfitci y1 x, level(90) estopts(degree(4))
```

Overlay a scatterplot

```
twoway fpfitci y1 x || scatter y x
```

As above, but use small circles as markers

```
twoway fpfitci y1 x || scatter y x, msymbol(o)
```

Display confidence limits as a pair of lines

```
twoway fpfitci y1 x, ciplot(rline)
```

As above, with an overlaid scatterplot and medium-thick line for predicted values

```
twoway fpfitci y1 x, ciplot(rline) clwidth(medthick) || scatter y1 x
```

Predictions from estimation command logit with binary dependent variable *y2*

```
twoway fpfitci y2 x, estcmd(logit)
```

Add “My Title” as the title of the graph

```
twoway fpfitci y1 x || scatter y x, title("My Title")
```

### Menu

Graphics > Twoway graph (scatter, line, etc.)
Syntax

```
twoway fpfitci yvar xvar [if] [in] [weight] [ , options ]
```

<table>
<thead>
<tr>
<th>options</th>
<th>Description</th>
</tr>
</thead>
<tbody>
<tr>
<td><code>fpfit_options</code></td>
<td>any of the options of <code>[G-2] graph twoway fpfit</code></td>
</tr>
<tr>
<td><code>level(#)</code></td>
<td>set confidence level; default is <code>level(95)</code></td>
</tr>
<tr>
<td><code>nofit</code></td>
<td>prevent plotting the prediction</td>
</tr>
<tr>
<td><code>fitplot(plottype)</code></td>
<td>how to plot fit; default is <code>fitplot(line)</code></td>
</tr>
<tr>
<td><code>ciplot(plottype)</code></td>
<td>how to plot CIs; default is <code>ciplot(rarea)</code></td>
</tr>
<tr>
<td><code>fcline_options</code></td>
<td>change look of predicted line</td>
</tr>
<tr>
<td><code>fitarea_options</code></td>
<td>change look of CI</td>
</tr>
<tr>
<td><code>axis_choice_options</code></td>
<td>associate plot with alternative axis</td>
</tr>
<tr>
<td><code>twoway_options</code></td>
<td>titles, legends, axes, added lines and text, by, regions, name, aspect ratio, etc.</td>
</tr>
</tbody>
</table>

Option `level()` is rightmost; `nofit`, `fitplot()`, and `ciplot()` are unique; see `[G-4] Concept: repeated options.`

`aweight`, `fweight`, and `pweight` are allowed. Weights, if specified, affect estimation but not how the weighted results are plotted. See `[U] 11.1.6 weight.`

Options

`fpfit_options` refers to any of the options of `[G-2] graph twoway fpfit`; see `[G-2] graph twoway fpfit.` These options are seldom specified.

`level(#)` specifies the confidence level, as a percentage, for the confidence intervals. The default is `level(95)` or as set by `set level`; see `[U] 20.8 Specifying the width of confidence intervals.`

`nofit` prevents the prediction from being plotted.

`fitplot(plottype)` is seldom specified. It specifies how the prediction is to be plotted. The default is `fitplot(line)`, meaning that the prediction will be plotted by `graph twoway line`. See `[G-2] graph twoway` for a list of `plottype` choices. You may choose any plottypes that expect one `y` variable and one `x` variable.

`ciplot(plottype)` specifies how the confidence interval is to be plotted. The default is `ciplot(rarea)`, meaning that the prediction will be plotted by `graph twoway rarea`.

A reasonable alternative is `ciplot(rline)`, which will substitute lines around the prediction for shading. See `[G-2] graph twoway` for a list of `plottype` choices. You may choose any plottypes that expect two `y` variables and one `x` variable.

`fcline_options` specify how the prediction line is rendered; see `[G-3] fcline_options`. If you specify `fitplot()`, then rather than using `fcline_options`, you should select options that affect the specified `plottype` from the options in `scatter`; see `[G-2] graph twoway scatter`.

`fitarea_options` specify how the confidence interval is rendered; see `[G-3] fitarea_options`. If you specify `ciplot()`, then rather than using `fitarea_options`, you should specify whatever is appropriate.

`axis_choice_options` associate the plot with a particular `y` or `x` axis on the graph; see `[G-3] axis_choice_options.`
Typical use

twoway fpfitci by default draws the confidence interval of the predicted mean:

```
. use https://www.stata-press.com/data/r16/auto
   (1978 Automobile Data)
. twoway fpfitci mpg weight
```
If you specify the `ciplot(rline)` option, the confidence interval will be designated by lines rather than shading:

```
twoway fpfitci mpg weight, ciplot(rline)
```

In the above graph, the shaded area corresponds to the 95% confidence interval for the mean.

**Advanced use**

`fpfitci` can be usefully overlaid with other plots:

```
use https://www.stata-press.com/data/r16/auto, clear
(1978 Automobile Data)

twoway fpfitci mpg weight || scatter mpg weight
```

In the above graph, the shaded area corresponds to the 95% confidence interval for the mean.
It is of great importance to note that we typed

```plaintext
    . twoway fpfitci ... || scatter ...
```

and not

```plaintext
    . twoway scatter ... || fpfitci ...
```

Had we drawn the scatter diagram first, the confidence interval would have covered up most of the points.

**Cautions**

Do not use `twoway fpfitci` when specifying the `axis_scale_options` 
`yscale(log)` or `xscale(log)` to create log scales. Typing

```plaintext
    . twoway fpfitci mpg weight || scatter mpg weight ||, xscale(log)
```

will produce a curve that will be fit from a fractional polynomial regression of `mpg` on `weight` rather than `log(weight)`.

See *Cautions* in [G-2] `graph twoway lfitci`.

**Use with by()**

`fpfitci` may be used with `by()` (as can all the `twoway` plot commands):

```plaintext
    . twoway fpfitci mpg weight || 
        scatter mpg weight ||, by(foreign, total row(1))
```

Also see

[G-2] `graph twoway lfitci` — Twoway linear prediction plots with CIs

[G-2] `graph twoway qfitci` — Twoway quadratic prediction plots with CIs

[G-2] `graph twoway fpfit` — Twoway fractional-polynomial prediction plots